# Copyright 2009 Google Inc.
#
# Licensed under the Apache License, Version 2.0 (the "License");
# you may not use this file except in compliance with the License.
# You may obtain a copy of the License at
#
#     http://www.apache.org/licenses/LICENSE-2.0
#
# Unless required by applicable law or agreed to in writing, software
# distributed under the License is distributed on an "AS IS" BASIS,
# WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
# See the License for the specific language governing permissions and
# limitations under the License.
#
# Author: Gunnar Ritter

# A Qt project file to build a command line application to
# execute the probes, passing the account data as options.

TEMPLATE=app

CONFIG += qt
CONFIG += debug_and_release
CONFIG += build_all
CONFIG += console

CONFIG(debug, debug|release) {
  TARGET = debug_diag
  DESTDIR = build/dbg
  MOC_DIR = build/dbg/obj
  RCC_DIR = build/dbg/obj
  OBJECTS_DIR = build/dbg/obj
} else {
  TARGET = diag
  DESTDIR = build/opt
  MOC_DIR = build/opt/obj
  RCC_DIR = built/opt/obj
  OBJECTS_DIR = build/opt/obj
}

QT += core network xml

INCLUDEPATH += ..

win32 {
  SOURCES += ../probes/localhost_win.cc
}

mac {
  SOURCES += ../probes/localhost_mac.cc
}

unix {
  SOURCES += ../probes/localhost_unix.cc
}

HEADERS += ../cmdline/cmdline.h \
           ../probes/executive.h \
           ../probes/imapprobe.h \
           ../probes/linereader.h \
           ../probes/localhost.h \
           ../probes/multiprobe.h \
           ../probes/popprobe.h \
           ../probes/probe.h \
           ../probes/smtpprobe.h \
           ../probes/ssl.h \
           ../ui/reportgen.h \
           ../ui/support.h \
           ../util/common.h \
           ../util/protocol.h \
           ../util/provider.h

SOURCES += ../cmdline/cmdline.cc \
           ../cmdline/main.cc \
           ../probes/executive.cc \
           ../probes/imapprobe.cc \
           ../probes/linereader.cc \
           ../probes/localhost.cc \
           ../probes/multiprobe.cc \
           ../probes/popprobe.cc \
           ../probes/probe.cc \
           ../probes/smtpprobe.cc \
           ../probes/ssl.cc \
           ../ui/reportgen.cc \
           ../util/common.cc \
           ../util/protocol.cc \
           ../util/provider.cc
